# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Mark Wagie <mark dot wagie at tutanota dot com>

pkgname=blanket
pkgver=0.6.0
pkgrel=2
pkgdesc='Improve focus and increase your productivity by listening to different sounds'
arch=('any')
url='https://github.com/rafaelmardojai/blanket'
license=('GPL3')
depends=('gst-plugins-bad-libs' 'gst-plugins-base' 'gst-plugins-good' 'gst-python' 'libadwaita' 'python-gobject')
makedepends=('appstream-glib' 'meson')
source=("https://github.com/rafaelmardojai/$pkgname/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('710a3b733da5eb349bf8dc3c37ef0afeb301c8299c22e7fad3c753576f33f7f4')

build() {
  arch-meson $pkgname-$pkgver build
  meson compile -C build
}

check() {
  meson test -C build --print-errorlogs
}

package() {
  DESTDIR="$pkgdir" meson install -C build
}
